

RDM 4.11 and DeployCenter 5.5

A White Paper On

Using IBM Remote Deployment Manager 4.11
With PowerQuest DeployCenter 5.5
To Deploy Windows Operating Systems,
Including Multiple Partitions

August, 2003
Updated October 2003

Table of Contents

<i>Introduction</i>	3
<i>Customizing RDM 4.11 (without Update 1) to Use ImageCenter</i>	4
<i>Customizing RDM 4.11 (with Update 1) to Use ImageCenter</i>	8
<i>Notices</i>	9

Introduction

IBM© Remote Deployment Manager (RDM) version 4.11 is capable of capturing the entire boot partition of a properly deployed “donor” System into a single, compressed “image” file and installing that “image” onto multiple “target” Systems configured with identical hardware. This makes for an efficient way to deploy the Microsoft™ Windows™ operating system onto multiple identical Systems. However, there are technical and contractual limitations in the technology used by RDM. These include being able to capture and deploy *only* the boot partition and having to deploy to boot partitions of identical size to the boot partition in the “donor” System.

The technology used in RDM for “cloning” is a customized, “lite” version of the ImageCenter engine from PowerQuest, Inc. This is based on the “full” ImageCenter engine that ships with PowerQuest’s DeployCenter product. It is possible for users to purchase PowerQuest DeployCenter and use the “full” ImageCenter engine with customized RDM tasks, getting around the limitations of the “lite” ImageCenter engine. Use of the “full” ImageCenter allows capturing of multiple partitions (for example, the Boot partition plus a Service partition and/or an Extended partition containing one or more logical drives). The example in this paper includes capturing and deploying the entire boot drive, whether it contains a single partition or multiple partitions.

The following instructions are based on the “full” version of the ImageCenter engine that is shipped with PowerQuest DeployCenter 5.5. Other versions may have different functionality, script commands, file names and/or file locations, possibly negating these instructions.

Although one of the features of RDM is to distribute software to systems across a LAN, the terms and conditions of the IBM International Program License Agreement for RDM do not grant any license to install, copy, or use any application software or operating system software not provided with RDM. This includes, but is not limited to, Windows 2000 Professional, Windows 2000 Server and Advanced Server, Windows XP Professional, Windows Server 2003, DOS, and Linux. Always ensure that you have obtained suitable licenses for any software you intend to use with RDM.

Customizing RDM 4.11 (without Update 1) to Use ImageCenter

The following instructions assume that IBM Remote Deployment Manager 4.11 (*without* Update 1) and PowerQuest DeployCenter 5.5 have been previously installed. Please substitute the path into which RDM has been installed on your server for **rdmRoot**. Note that these instructions will change the default behavior of cloning Windows Systems under RDM. You may wish to keep backup copies of the files to be edited and restore them to return to the default RDM behavior.

1. Copy the following DeployCenter files (which constitute the ImageCenter engine) from the DeployCenter folder (the default of which is **Program Files\PowerQuest\DeployCenter 5.5**) to the **rdmRoot\repository\environment\dos** folder:
 - PQImgCtr.exe**
 - PQImgCtr.ovl**
 - PQImgCtr.pqg**
 - PQDplCtr.rtc**
2. Edit **PQSTORE.SCR** in the **rdmRoot\local\env\71c\capture** folder to have the lines shown in the *PQSTORE.SCR* section below
3. Edit **CAPTURE.BAT** in the **rdmRoot\local\env\71c\capture** folder to have the lines shown in the *CAPTURE.BAT* section below
4. Edit **PQRESTR.SCR** in the **rdmRoot\local\env\71c\deploy** folder to have the lines shown in the *PQRESTR.SCR* section below
5. Edit **DEPLOY.BAT** in the **rdmRoot\local\env\71c\deploy** folder to have the lines shown in the *DEPLOY.BAT* section below
6. Run the **MKIMAGES.BAT** file in the **rdmRoot\local\env** folder
7. Following the instructions in the *Scenario using PQIDplyZ.exe and sysprep* section of the *Remote Deployment Manager 4.11 Operations Guide*, prepare a “donor” machine
8. Following the instructions in the *Schedule a Get Donor task* section of the *Remote Deployment Manager 4.11 Operations Guide*, create an “image” of the donor machine
9. Following the instructions in the *Windows Clone Install* section of the *Remote Deployment Manager 4.11 Operations Guide*, define a Windows Clone Install task
10. Using one of the operations defined in the *Running Tasks on Systems* section of the *Remote Deployment Manager 4.11 Operations Guide*, perform a Windows Clone Install task on one or more Systems.

Note that in the following sections, the “␣” symbol is used to indicate line breaks that would *not* be present in the actual files but appear in this document to fit the formatted pages. Please enter the text in all lines starting with the “␣” symbol as extensions of the previous line.

PQSTORE.SCR

```
SELECT DRIVE 1
SELECT PARTITION ALL
SET DESCRIPTION ""
STORE WITH COMPRESSION HIGH
```

Note that this script can be modified to select only the partitions you wish to capture. Please refer to your PowerQuest DeployCenter documentation on scripting syntax.

CAPTURE.BAT

```
REM *****
REM * CAPTURE.BAT
REM * Copy client hard disk image to D-Server
REM *****
@ECHO OFF
@ECHO Running CAPTURE.BAT to create clone image from %CLIENT_IP%.
@DEL A:\SCRUB3.EXE
@SET RDSTATUS=RDMPQC000I
@ECHO MTFP copying over PQImgCtr files...
@MTFTP GET %SERVER_IP% environment\dos\pqimgctr.exe A:\pqimgctr.exe
@if ERRORLEVEL 1 GOTO DO_LITE
@DEL A:\MAKEPQS.COM
@MTFTP GET %SERVER_IP% environment\dos\pqdplctr.rtc A:\pqdplctr.rtc
@if ERRORLEVEL 1 GOTO TFTPERR
@MTFTP GET %SERVER_IP% environment\dos\pqimgctr.ovl A:\pqimgctr.ovl
@if ERRORLEVEL 1 GOTO TFTPERR
@MTFTP GET %SERVER_IP% environment\dos\pqimgctr.pqg A:\pqimgctr.pqg
@if ERRORLEVEL 1 GOTO TFTPERR
@ECHO.
@ECHO Using DeployCenter version...
@PQImgCtr /IMG=\\.\imgsrv\%SERVER_IP%\temp\%bootmac%\clone.pqi
␣ /CMD=A:\CAPTURE\PQSTORE.SCR /IFC /NOCDR /NMD /NUI /NRB
@GOTO RETURN
:DO_LITE
@ECHO MTFP copying over PQIDplyZ files...
@MTFTP GET %SERVER_IP% environment\dos\pqidplyz.exe A:\PQIDplyZ.exe
@if ERRORLEVEL 1 GOTO TFTPERR
@MTFTP GET %SERVER_IP% environment\dos\pqidplyz.rtc A:\pqidplyz.rtc
@if ERRORLEVEL 1 GOTO TFTPERR
@ECHO.
@ECHO Using RDM Lite version...
@MAKEPQS.COM
@PQIDplyZ /IMG=\\.\imgsrv\%SERVER_IP%\temp\%bootmac%\clone.pqi
␣ /CMD=PQSTORE.SCR /IFC
@GOTO RETURN
:RETURN
@SET RDSTATUS=RDMPQC001I
@if ERRORLEVEL 2 goto ERROR
@if ERRORLEVEL 1 SET RDSTATUS=RDMPQC001w
@GOTO END
:TFTPERR
```

```

@CALL MFTTPRC.BAT
@goto END
:ERROR
@CALL PQERROR.BAT
:END
@ECHO Exiting Capture.bat.

```

Note that this batch file allows for the presence or absence of the “full” ImageCenter files. If they are present, they will be used along with the customized script files. If they are missing, the “lite” version of the ImageCenter engine will be used. This batch file can be edited to simply “fail out” if the “full” ImageCenter files are not found. To do so, replace these lines:

```
@IF ERRORLEVEL 1 GOTO DO_LITE
```

With these lines:

```
@IF ERRORLEVEL 1 GOTO TFTPERR
```

PQRESTR.SCR

```

SELECT DRIVE 1
DELETE ALL
SELECT FREESPACE FIRST
SELECT IMAGE ALL
RESIZE IMAGE NO
RESTORE

```

Note that this script can be modified to select only the partitions you wish to capture. Note also that advanced features, such as proportional resizing of selected partitions, can be performed by adding/editing the proper lines in this script file. Please refer to your PowerQuest DeployCenter documentation on scripting syntax.

DEPLOY.BAT

```

REM *****
REM * DEPLOY.BAT
REM * Copy PowerQuest image from D-Server to client
REM *****
@ECHO OFF
@ECHO Running DEPLOY.BAT to deploy images stored within %CloneFolder%
↳ to client.
@REM Wipe the boot record of the first disk in case disk is corrupt
@MFTFP GET %SERVER_IP% environment\dos\scrub3.exe A:\scrub3.exe
@A:\SCRUB3 /D=1 /L=1 /S=NO
@DEL A:\SCRUB3.EXE
@SET RDSTATUS=RDMPQC002I
@ECHO MFTFP copying over PQImgCtr files...
@MFTFP GET %SERVER_IP% environment\dos\pqimgctr.exe A:\pqimgctr.exe
@IF ERRORLEVEL 1 GOTO DO_LITE
@MFTFP GET %SERVER_IP% environment\dos\pqdplctr.rtc A:\pqdplctr.rtc
@IF ERRORLEVEL 1 GOTO TFTPERR
@MFTFP GET %SERVER_IP% environment\dos\pqimgctr.ovl A:\pqimgctr.ovl
@IF ERRORLEVEL 1 GOTO TFTPERR
@MFTFP GET %SERVER_IP% environment\dos\pqimgctr.pqg A:\pqimgctr.pqg

```

```

@if ERRORLEVEL 1 GOTO TFTPERR
@ECHO.
@ECHO Using DeployCenter version...
@PQImgCtr
  /IMG=\\.\imgsrv\%SERVER_IP%\repository\image\%CloneFolder%\%CloneFileName%
  /CMD=deploy\PQRESTR.SCR /IFC /NOCDR /NUI /NRB
@goto RETURN
:DO_LITE
@ECHO MTFP copying over PQIDplyZ files...
@MTFTP GET %SERVER_IP% environment\dos\pqidplyz.exe A:\PQIDplyZ.exe
@if ERRORLEVEL 1 GOTO TFTPERR
@MTFTP GET %SERVER_IP% environment\dos\pqidplyz.rtc A:\pqidplyz.rtc
@if ERRORLEVEL 1 GOTO TFTPERR
@ECHO.
@ECHO Using RDM Lite version...
@PQIDplyZ
  /IMG=\\.\imgsrv\%SERVER_IP%\repository\image\%CloneFolder%\%CloneFileName%
  /CMD=deploy\PQRESTR.SCR /IFC
@goto RETURN
:RETURN
@SET RDSTATUS=RDMPC003I
@if ERRORLEVEL 2 GOTO ERROR
@if ERRORLEVEL 1 SET RDSTATUS=RDMPC001w
@goto END
:TFTPERR
@CALL MTFTPRC.BAT
@goto END
:ERROR
@CALL PQERROR.BAT
:END
@ECHO Exiting DEPLOY.BAT.

```

Note that this batch file allows for the presence or absence of the “full” ImageCenter files. If they are present, they will be used along with the customized script files. If they are missing, the “lite” version of the ImageCenter engine will be used. This batch file can be edited to simply “fail out” if the “full” ImageCenter files are not found. To do so, replace these lines:

```
@IF ERRORLEVEL 1 GOTO DO_LITE
```

With these lines:

```
@IF ERRORLEVEL 1 GOTO TFTPERR
```

Customizing RDM 4.11 (with Update 1) to Use ImageCenter

The following instructions assume that IBM Remote Deployment Manager 4.11, Update 1 for RDM 4.11 and PowerQuest DeployCenter 5.5 have been previously installed. Please substitute the path into which RDM has been installed on your server for **rdmRoot**. Note that these instructions will change the default behavior of cloning Windows Systems under RDM. You may wish to keep a backup copy of the file to be edited and restore it to return to the default RDM behavior.

1. Copy the following DeployCenter files (which constitute the ImageCenter engine) from the DeployCenter folder (the default is **Program Files\PowerQuest\DeployCenter 5.5**) to the **rdmRoot\local\env\71c** folder:
PQImgCtr.exe
PQImgCtr.ovl
PQImgCtr.pqg
PQDplCtr.rtc
2. Edit **PQSTORE.SCR** in the **rdmRoot\local\env\71c\capture** folder to have the lines shown in the **PQSTORE.SCR** section below
3. Run the **MKIMAGES.BAT** file in the **rdmRoot\local\env** folder
4. Following the instructions in the *Scenario using PQIDplyZ.exe and sysprep* section of the *Remote Deployment Manager 4.11 Operations Guide*, prepare a “donor” machine
5. Following the instructions in the *Schedule a Get Donor task* section of the *Remote Deployment Manager 4.11 Operations Guide*, create an “image” of the donor machine
6. Following the instructions in the *Windows Clone Install* section of the *Remote Deployment Manager 4.11 Operations Guide*, define a Windows Clone Install task
7. Using one of the operations defined in the *Running Tasks on Systems* section of the *Remote Deployment Manager 4.11 Operations Guide*, perform a Windows Clone Install task on one or more Systems.

PQSTORE.SCR

```
SELECT DRIVE 1
SELECT PARTITION ALL
SET DESCRIPTION ""
STORE WITH COMPRESSION HIGH
```

Note that this script can be modified to select only the partitions you wish to capture. Please refer to your PowerQuest DeployCenter documentation on scripting syntax.

Notices

IBM is a trademark of International Business Machines Corporation in the United States, other countries, or both:

Microsoft and Windows are trademarks of Microsoft Corporation in the United States, other countries, or both.

Other company, product, or service marks may be trademarks or service marks of others.